Effect Size Estimates for One-Way Repeated Measures ANOVA

These are usually proportion of variance estimates, despite the assorted problems with such estimates. Two choice are eta-squared (aka semipartial eta-squared) and partial eta-squared. The former includes, in the denominator, all the variance in the outcome variable Y. The latter excludes from the denominator variance due to any other factor (predictor, X) in the model. Accordingly, the latter will produce a larger estimate of the proportion of variance explained. In factorial designs it is possible for the sum of partial eta-squared values, across factors to exceed 100%. This makes me uncomfortable with partial eta-squared. The only cases where I think one might justify using partial eta-squared is when the factors being excluded from the denominator are factors that are unnatural in the sense that they exist only in the lab, not in the real word.

I shall illustrate the computation of eta-squared and partial eta-squared using the data in Table 14 of the 8th edition of David Howell's *Statistical Methods for Psychology*. Here are the data and SAS code to do the one-way repeated measures ANOVA. Do note that the data are in the "multivariate" format, with each line of data containing all of the scores for one subject (one score for each of the five weeks).

```
data ache; input week1-week5;
MeanSubj=Mean(of week1-week5); cards;
21 22  8  6  6
20 19 10  4  4
17 15  5  4  5
25 30 13 12 17
30 27 13  8  6
19 27  8  7  4
26 16  5  2  5
17 18  8  1  5
26 24 14  8  9
proc anova; model week1-week5= / nouni; repeated week 5 / nom; run; quit;
```

Here are parts of the resulting output:

Source	DF	Anova SS	Mean Square	F Value	Pr > F	Adj Pr > F	
						G - G	H - F
week	4	2449.200000	612.300000	85.04	<.0001	<.0001	<.0001
Error(week)	32	230.400000	7.200000				

Greenhouse-Geisser Epsilon	0.6845
Huynh-Feldt Epsilon	1.0756

Notice that SAS gives no estimates of effect size. We can, however, easily calculate the partial eta-squared: $\eta_p^2 = \frac{SS_{Week}}{SS_{Week} + SS_{Error\,(week)}} = \frac{2449.2}{2449.2 + 230.4} = .914$. Please note that the variation due to individual differences, the $SS_{Subjects}$ has been excluded from the denominator.

Next I execute this code:

```
proc means CSS; Var MeanSubj; run;
```

The MEANS Procedure

Analysis Variable: MeanSubj
Corrected SS
97.3422222

In the data step I created, for each subject, the mean score across five weeks. Proc Means finds the corrected sum of squares for those means. When I multiply that by the number of levels of the repeated dimension I get the $SS_{Subjects}$ = 5(97.3422222) = 486.71.

I can now calculate the $SS_{Total} = SS_{Subjects} + SS_{Weeks} + SS_{Error(weeks)}$ = 486.71 + 2449.2 + 230.4 = 3166.31.

Now I can calculate eta-squared: $\eta^2 = \frac{SS_{Week}}{SS_{Total}} = \frac{2449.2}{3166.31} = .7735$. Notice that the eta-squared is less than the partial eta-squared because the eta-squared includes in the denominator variance due to individual differences among subjects but the partial eta-squared does not. Since individual differences among subjects exists in the natural world, I think it inappropriate to exclude that variance from the denominator.

There is another more direct way to get the total sum of squares. Here is the SAS code:

```
data total; input Y @@; cards;
21 22  8  6  6
20 19 10  4  4
17 15  5  4  5
25 30 13 12 17
30 27 13  8  6
19 27  8  7  4
26 16  5  2  5
17 18  8  1  5
26 24 14  8  9
proc means css; var Y; run;
```

The code reads in all scores from all subjects as a single variable. Proc Means is then employed to get the total sum of squares.

The MEANS Procedure

Analysis Variable: Y
Corrected SS
3166.31

If the data are in “univariate” layout, the EFFECTSIZE command can be used to get eta-squared, omega squared, partial eta-squared, and partial omega-squared.

```
data headache; input subject week duration; cards;
1 1 21
1 2 22
```

```
1 3 8
1 4 6
1 5 6
2 1 20
2 2 19
2 3 10
2 4 4
2 5 4
3 1 17
3 2 15
3 3 5
3 4 4
3 5 5
4 1 25
4 2 30
4 3 13
4 4 12
4 5 17
5 1 30
5 2 27
5 3 13
5 4 8
5 5 6
6 1 19
6 2 27
6 3 8
6 4 7
6 5 4
7 1 26
7 2 16
7 3 5
7 4 2
7 5 5
8 1 17
8 2 18
8 3 8
8 4 1
8 5 5
9 1 26
9 2 24
9 3 14
9 4 8
9 5 9
proc GLM; class subject week; model duration = subject week / SS1 effectsize; run; quit;
```

The GLM Procedure

Dependent Variable: duration

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	12	2935.911111	244.659259	33.98	<.0001
Error	32	230.400000	7.200000		
Corrected Total	44	3166.311111			

R-Square	Coeff Var	Root MSE	duration Mean
0.927234	20.25968	2.683282	13.24444

Source	DF	Type I SS	Mean Square	F Value	Pr > F	Total Variation Accounted For				Partial Variation Accounted For			
						Semipartial Eta-Square	Semipartial Omega-Square	Conservative 95% Confidence Limits		Partial Eta-Square	Partial Omega-Square	95% Confidence Limits	
subject	8	486.711111	60.838889	8.45	<.0001	0.1537	0.1352	0.0000	0.2065	0.6787	0.5698	0.3357	0.7159
week	4	2449.200000	612.300000	85.04	<.0001	0.7735	0.7627	0.6033	0.8290	0.9140	0.8819	0.7997	0.9232

Now the same analysis done with SPSS.

Tests of Within-Subjects Effects

Measure: MEASURE_1

Source		Type III Sum of Squares	df	Mean Square	F	Sig.	Partial Eta Squared
week	Sphericity Assumed	2449.200	4	612.300	85.042	.000	.914
	Greenhouse-Geisser	2449.200	2.738	894.577	85.042	.000	.914
	Huynh-Feldt	2449.200	4.000	612.300	85.042	.000	.914
	Lower-bound	2449.200	1.000	2449.200	85.042	.000	.914
Error(week)	Sphericity Assumed	230.400	32	7.200			
	Greenhouse-Geisser	230.400	21.903	10.519			
	Huynh-Feldt	230.400	32.000	7.200			
	Lower-bound	230.400	8.000	28.800			

I created the subject means this way:



Now a one-way ANOVA predicting subject means from subject numbers



ANOVA

SubjMean

	Sum of Squares	df	Mean Square	F	Sig.
Between Groups	97.342	8	12.168	.	.
Within Groups	.000	0	.		
Total	97.342	8			

Multiply the sum of squares by the number of levels of the repeated measures factor to get the subjects sum of squares: 5(97.342) = 486.71.

Bruce Weaver showed me a much easier way to get the subjects sum of squares – just look at the Tests of Between Subjects Effects table (which I am conditioned to ignore in a repeated measures design) – the *SS* for "Error" is the subjects sum of squares.

Tests of Between-Subjects Effects

Measure: MEASURE_1

Transformed Variable: Average

Source	Type III Sum of Squares	df	Mean Square	F	Sig.
Intercept	7893.689	1	7893.689	129.747	.000
Error	486.711	8	60.839		

The total sum of squares in the sum of the sums of squares for week, subjects, and error = 2449.2 + 486.71 + 230.4 = 3166.31. Eta-squared = 2449.2/3166.31 = .7735.

Bruce (aka SPSS Wizard) also provided the syntax to do all this without a calculator.

Karl L. Wuensch, June, 2017